

Elektronski fakultet u Nišu
Katedra za mikroelektroniku

PROJEKTOVANJE MIKROELEKTRONSKIH KOMPONENATA


LABORATORIJSKE VEŽBE

NEDA BRANKOVIĆ

DIODA TEHNOLOŠKI NIZ ATHENA

1. Domen simulacije

2. Definisiranje mreže



`go athena`

```
line x loc=0.0 spacing=0.5
line x loc=2.5 spacing=0.025
line x loc=5 spacing=0.5
line y loc=0.0 spacing=0.01
line y loc=5 spacing=0.5
```

neuniformna mreža - nije jednake gustine (previše gusta mreža - dugo trajanje simulacije, precizna simulacija; previše retka mreža - brzo izvršenje simulacije, nisu zahvaćena značajna mesta u strukturi).

Domen: pločica je dimenzija 5x5 μm

je komentar - ne kompajlira se!

Red veličine: μm

Treća dimenzija **z** se podrazumeva da je 1 μm iako se ne definiše.

3. Definisanje supstrata: `init`

`Supstrat` - osnova na kojoj se odvijaju svi ostali tehnološki procesi koji slede.


```
init silicon <100> c.phosphorus=1e15 two.d
```

Početni materijal: `Si`

`<100>` ili `orientation=100`: kristalografska orijentacija- Milerovi indeksi

Dopiranje se definiše pomoću koncentracije određene primese: fosfor - $N_D=10^{15} \text{ cm}^{-3}$

Prikaz: `2D`



4. Oksidacija silicijuma: **diffuse**

Oksidacija - tehnološki proces formiranja sloja silicijum-dioksida na površini silicijuma. Može biti termička (800-1000°C) i hemijska (200-600°C).

```
diffuse time=40 temp=1100 weto2
```

```
structure outfile=prvadif.str
```

```
tonyplot prvadif.str
```

Formiranje silicijum-dioksida na površini neophodno je za naredne tehnološke procese koji slede. Oksid ima ulogu maske.


Odvijanje osidacije: 40 minuta na 1100°C u ambijentu bogatom vodenom parom - weto2

Prilikom formiranja sloja silicijum-dioksida, deo silicijumskog supstrata koji reaguje sa kiseonikom se izgubi.

```
structure outfile=imefajla.str - čuvanje  
strukture u izlazni fajl, ekstenzija .str
```

```
tonyplot imefajla.str - prikaz rezultata
```

SiO_2 je dielektrik!




5. Nagrizanje oksida: **etch**

Nagrizanje je tehnološki proces otklanjanja/ecovanja/nagrizanja nekog materijala sa površine silicijumskog supstrata.


```
etch oxide start x=2 y=-5
etch continue x=2 y=1
etch continue x=3 y=1
etch done x=3 y=-5
```

```
structure outfile=nagr.str
tonyplot nagr.str
```

U liniji koda u kojoj se definišu koordinate prve tačke poligona koji se ecuje, neophodno je navesti materijal od kojeg je taj poligon napravljen.



Navođenje koordinati za delove koje treba ecovati- neophodno je da se koordinate nadovezuju, odnosno da zatvaraju jedan pun krug!





6. Jonska implantacija - **implant** i 7. difuzija - **diffuse**

Jonska implantacija je proces ubacivanja primesa (jona) u silicijum. Kontrolišući parametri: doza i energija.



Difuzija je proces dopiranja silicijuma, pri čemu se vrši preraspodela primesa unetih procesom jonske implantacije u supstratu. Odvija na visokim temperaturama.

Implantacija: model simulacije - **gauss**; proces se vrši borom doze $5e14$, sa energijom 60 keV i nagibnim uglom 0° (direktno odozgo)

Difuzija: oksidacija (odžarivanje) - u trajanju od 45 minuta, na temperaturi od 1050°C , u ambijentu bogatom kiseonikom.

**implant gauss boron dose=5e14 energy=60 tilt=0
diffuse time=45 temp=1050 dryo2**

**structure outfile=implant.str
tonyplot implant.str**



8. Otvor za kontakt anode

```
etch oxide start x=2.2 y=-5
etch continue x=2.2 y=1
etch continue x=2.8 y=1
etch done x=2.8 y=-5
```

→ Ecuje se oksid koji je narastao tokom izvođenja procesa difuzije u prethodnom koraku.

9. Depozicija aluminijuma

Depozicija je proces nanošenja nekog materijala na površinu silicijumske pločice.


```
deposit aluminum thick=1 div=20
```

```
structure outfile=conanode.str
tonyplot conanode.str
```

Obavezno navesti materijal koji treba da se deponuje.

debljina materijala (μm): **thick**

div (division): parametar za formiranje mreže



10. Uklanjanje viška aluminijuma

```
etch aluminum p1.x=1.5 left
etch aluminum p1.x=3.5 right
```


```
structure outfile=alum1.str
tonyplot alum1.str
```

Aluminijum ostaje samo na mestu gde će biti kontakt za anodu!

Uklanjanje aluminijuma od pozicije:

p1.x=1.5 left - levo od 1.5 μm

p1.x=3.5 right - desno od 3.5 μm





11. Definisanje elektroda: **electrode**

```
electrode x=2.5 y=0 name=anode
electrode backside name=cathode
```

```
structure outfile=dioda_final.str
tonyplot dioda_final.str
quit
```

x=2.5 y=0 name=anode: navođenje koordinata i definisanje imena kontakta
backside name=cathode: cela donja strana supstrata je drugi kontakt

Anoda p oblast, katoda n oblast!



SIMULACIJA ELEKTRIČNIH KARAKTERISTIKA DIODE

ATLAS

Za simulaciju električnih karakteristika koristi se potprogram Atlas.

1. Električne karakteristike diode pri direktnoj polarizaciji

go atlas

contact name=anode

models bipolar temp=300

method newton

method: definisanje matematičkih metoda za rešavanje jednačina

newton: Njutnov metod

- Na početku se uvek navode osnovne naredbe koje definišu simulaciju električnih karakteristika.

contact name=anode: definisanje kontakta na koji se dovodi napon

models: model koji se koristi u simulaciji

bipolar: dioda je PN spoj - u provođenju struje učestvuju i elektroni i šupljine, a model u proračunu struja uzima u obzir obe vrste naleketrisanja

temp=300: temperatura na kojoj dioda radi u Kelvinima

```
solve init
log outfile=nasadioda.log
solve vanode=0 vstep=0.1 vfinal=1 name=anode
```

```
tonyplot nasadioda.log
quit
```

log outfile=nazivfajla.log: otvaranje izlaznog fajla u kome se čuvaju numerički rezultati **naredne solve naredbe**

Naredbom **solve** napona na anodi se menja od 0 V do 1 V sa korakom 0.1 V.

vanode=0: **naponska pobuda, početna vrednost napona**

vstep=0.1: **korak promene napona**

vfinal=1: **krajnja vrednost napona**

name=anode: **napon se menja na anodi**

- **Pobuda može biti i strujna!**
- **Tekuća solve naredba uzima zadnje rešenje iz prethodne solve naredbe!**

- **U svakoj simulaciji se prvo traži početno rešenje!**

solve: naredba za rešavanje numeričkih jednačina

init: naredba za nalaženje početnog rešenja

- **Rezultati ove naredbe se ne čuvaju!**

Naredbom solve init simulator na svim izvodima postavlja 0 i vrši računanje kako bi dobio početna rešenja.

- **Pre svake nove solve naredbe obavezno ide naredba solve init!**

- Analiza izlaznog fajla


Nakon odrađene simulacije, u izlaznom prozoru mogu se videti proračuni i karakteristike za napravljenu komponentu. Brisanje izlaznog prozora: *File - Clear Output Win ili donji levi ugao na ekranu (Clear)*.

```
Reading MASTER format file C:\Users\Public\  
Read 1787 nodes.  
Read 3394 triangles.  
Read 5 regions.  
Read 2 electrodes.  
3731509 net bytes allocated during read.
```



- broj učitanih čvorova iz mreže
- broj napravljenih trouglova u mreži formiranih iz preseka u mreži
- broj regiona: silicijum, oksid sa leve strane, oksid sa desne strane, anoda, katoda
- broj elektroda: anoda i katoda
- broj bajtova lociranih u memoriji prilikom iščitavanja

- koordinate elektroda
- analiza mreže: broj tupih uglova
- redom opis regiona
- rešenje solve init naredbe - 12 iteracija za dobijanje početnog rešenja
- rešenje solve naredbe: V_a - napon na anodi, J_n - gustina struje elektrona, J_p - gustina struje šupljina; redom preračunava: za $V_a=0$ V, $V_a=0.1$ V...
- struje su izražene u A/ μ m






```
init      psi      psi  
direct    x        rhs  
  i      j  m  -5.00* -26.0*  
-----  
  1      1  G  -0.563 -14.99  
  1      2  G  -0.614 -15.03  
  1      3  G  -0.753 -15.08  
  1      4  G  -0.829 -15.14  
  1      5  G  -0.699 -15.20  
  1      6  G  -0.778 -15.27  
  1      7  G  -0.845 -15.37  
  1      8  G  -0.903 -15.48  
  1      9  G  -0.954 -15.64  
  1     10  G  -1.000 -15.89  
  1     11  G  -1.677 -16.65  
  1     12  G  -15.2* -30.3*
```

DIODE - Električne karakteristike

U pitanju je silicijumska ispravljačka dioda, tako da se pri direktnoj polarizaciji očekuje da napon vođenja diode bude 0.7 V.

dioda: 5x5x1 -
struja u μA



Nakon što su dobijeni rezultati simulacije, sa karakteristike se može očitati da je napon vođenja diode 0.7 V.

2. Simulacija električnih karakteristika diode pri inverznoj polarizaciji

```

solve init
solve vanode=0 vstep=-0.25 vfinal=-10 name=anode ←
log outfile=nasadioda1.log
solve vanode=-10 vstep=0.1 vfinal=1 name=anode



tonyplot nasadioda1.log
    
```

Razlog uvođenja *dodatne/pomoćne* solve naredbe je preveliki razmak između 0 V (solve init) i krajnje vrednosti napona.

`log outfile` naredba *preskače* se između solve init naredbe i dodatne/pomoćne solve naredbe, jer nije potrebno čuvati rezultate pomoćne naredbe, ali je *obavezna* između dodatne/pomoćne solve naredbe i glavne solve naredbe.

Napomena: samo rezultati glavne solve naredbe se čuvaju!

DIODE - Električne karakteristike



```
save outfile=nasadioda.str  
tonyplot nasadioda.str
```

Vrednost napona pri kojoj dolazi do proboja ne može se videti zbog tipa modela (bipolar) koji se koristi u simulaciji!

```
save outfile=nazivfajla.str: služi za čuvanje rešenja jednačina  
rezličitih parametara u svakom čvoru mreže.
```

***.str** - strukturni fajl


***.log** - fajl koji sadrži numerička rešenja jednačina

1. Raspodela struje unutar strukture

Nakon naredbe save, rešenja jednačina različitih parametara sačuvani su u svakom čvoru mreže, tako da je došlo do modifikacija i u samoj strukturi simulirane komponente.

- Tonyplot (desni klik) – **Display – Contours**
- Dugme **Define – Contours**, otvara se novi prozor **Tonyplot: Contours**
- **Quantity – Total Current Density**

Struja se gomila u dve tačke (svi elektroni i šupljine koji učestvuju u provođenju struje biće nagomilani u tim tačkama) zbog kontaktne otpornosti diode!




2. Raspodela potencijala unutar komponente




- Tonyplot (desni klik) - **Display** - **Contours**
- Dugme **Define** - **Contours**, otvara se novi prozor **Tonyplot: Contours**
- **Quantity** - **Potential**

Da bi se posmatrale raspodele različitih parametara unutar diode kada je napon na anodi 0.7 V (kada dioda provede), potrebno je u solve naredbi definisati $v_{final}=0.7$:

```
solve vanode=0 vstep=0.1 vfinal=0.7 name=anode
```



Simulacija proboja diode



go atlas

```
contact name=anode
models bipolar temp=300
impact selb
method newton climit=1e-4
```

PROBOJ – dovođenjem napona inverzne polarizacije dolazi do povećanja širine osiromašene oblasti, odnosno lavinske multiplikacije nosilaca (nekontrolisanog povećanja struje usled velike kinetičke energije elektrona). Proboj se dešava odjednom, a da bi se videlo koleno karakteristike treba „usitniti“ korake pri rešavanju jednačina Njutnovim metodom.

impact selb – uključivanje Zelbeherovog modela

climit u naredbi method – koristi se za usitnjavanje koraka


```
solve init  
solve vanode=0 vstep=0.1 vfinal=1 name=anode  
log outfile=nasadioda.log  
solve vanode=1 vstep=-0.5 vfinal=-20 name=anode ←  
solve vanode=-20 vstep=-0.25 vfinal=-40 name=anode ←
```

Rezultati obe solve naredbe čuvaju se u outfile=nasadioda.log

```
tonyplot nasadioda.log
```

```
quit
```

- Da bi prikazali napon proboja, u solve naredbi napon na anodi se mora menjati od pozitivnih ka negativnim vrednostima!
- Da bi ubrzali simulaciju koriste se dve **solve** naredbe sa različitim korakom (**vstep**)!
- Rezultati simulacije pokazuju da je probojni napon veći od -40 V.




Promena vrednosti napona proboja:

1. duža (kraća) difuzija - primese prodiru dublje (pliće) u supstrat - dublji (plići) PN spoj - veći (manji) probojni napon;
2. manja (veća) inicijalna koncentracija supstrata - dublji (plići) PN spoj - veći (manji) probojni napon.

Promena u kodu:

```
init silicon <100> c.phosphorus=1e18 two.d
```



U nekom trenutku, dok izvršenje simulacije još uvek traje, Decbuild javlja grešku.

```
Warning: Convergence problem. Taking smaller bias  
step(s). Bias step reduced 4 times.
```

```
Warning: Newton algorithm did not converge in      25 iterations.
```

```
Warning: Bias step cut back more than 4  
times. Cannot trap.
```

Kada je simulacija završena, u izlaznom prozoru naći će se poruka: *Newton algorithm did not converge in 25 iterations.*

Simulator je usitnjavao korake, ali nije uspeo da pronade rešenje u 25 iteracija. To je dokaz da je došlo do proboja!


DIODE - Simulacija proboja diode

- *Tonyplot* se pokrene „ručno“ kako bi se videli rezultati simulacije. Opcijom *Open* učitava se poslednji fajl `nasadioda.log`.
- Sa grafika se može očitati da je vrednost napona pri kojoj nastupa proboj **-9.741 V**.

```
Warning: Convergence problem. Taking smaller bias  
step(s). Bias step reduced 4 times.
```

```
Obtaining static solution:
```

```
V( anode ) = -9.7412
```



DIODE - Simulacija proboja diode

- Da bi se prikazala raspodela struje pri proboju u strukturi treba napraviti i prikazati strukturni fajl, pri čemu napon **vfinal** treba postaviti blizu vrednosti proboja, kako bi podaci bili upisani. Ukoliko se ne izvrše ove modifikacije, doći će do prekida programa zbog problema konvergencije i podaci neće biti upisani u fajl.

```
solve vanode=1 vstep=-0.5 vfinal=-9.5 name=anode
```


```
save outfile=nasadioda.str
```

```
tonyplot nasadioda.log
```

```
tonyplot nasadioda.str
```

```
quit
```

- Tonyplot (desni klik) - **Display - Contours**
- Dugme **Define - Contours**, otvara se novi prozor **Tonyplot: Contours**
- **Quantity - e- Current Density**



Simulacija temperaturnih karakteristika diode


1. Ponoviti simulacije za različite temperature (komentarisati sve naredbe kojima se simulirao proboj diode).
2. Dati smisljena imena log fajlovima: **nasadioda_300.log**, **nasadioda_370.log**, **nasadioda_400.log**.
3. Naredbu za pozivanje **tonyplot**-a komentarisati **#** kako se ne bi otvarale sve zavisnosti posebno.

temp=300 K

temp=370 K

temp=400 K

<pre>go atlas contact name=anode models bipolar temp=300 method newton solve init log outfile=nasadioda300.log solve vanode=0 vstep=0.1 vfinal=1 name=anode quit</pre>	<pre>go atlas contact name=anode models bipolar temp=370 method newton solve init log outfile=nasadioda370.log solve vanode=0 vstep=0.1 vfinal=1 name=anode quit</pre>	<pre>go atlas contact name=anode models bipolar temp=400 method newton solve init log outfile=nasadioda400.log solve vanode=0 vstep=0.1 vfinal=1 name=anode quit</pre>
---	---	---



- Tonyplot („ručno“), iz padajućeg menija *Action-Overlay Plot* (da bi se karakteristike za sve tri različite temperature otvorile na istom grafiku), selektovati sva tri fajla.
- Može se zaključiti da se **sa povećanjem temperature smanjuje napon vođenja diode** - dioda se može koristiti kao senzor temperature, jer je njen napon vođenja diode linearna zavisnost $V_D=f(T)$, ali je potreban precizan strujni izvor kojim će se dioda napajati.